

OPENMIND®

MICROSOFT OFFICE 2007


EXCEL W FIRMIE®
ZAAWANSOWANY

ĆWICZENIE
ROZLICZENIE DELEGACJI

OPENMIND, 81-405 Gdynia, Ujejskiego 18
tel. (58) 661 0191, faks (58) 661 0192
email: openmind@openmind.pl, www.openmind.pl

ZAŁOŻENIA ĆWICZENIA

OPIS:

Jesteś odpowiedzialny za rozliczenie delegacji pracowników za pierwszy kwartał bieżącego roku. Do dyspozycji masz listę delegacji, dane na temat odległości do każdej lokalizacji, informacje na temat kwoty diety krajowej oraz kwot diet do każdego kraju a także aktualne kursy walut NBP. Korzystając z tych informacji musisz obliczyć kwoty wypłacane pracownikom.

Zasady rozliczania delegacji:

- Transport odbywa się wyłącznie samochodem, zwrot kosztów według stawki za jeden kilometr równej 0,8358.
- Delegacje zagraniczne rozliczane według właściwych stawek diety, przeliczane na PLN po aktualnym (z dnia wyjazdu) kursie NBP.
- Stawka diety za delegację krajową równa 23 PLN.
- Za każdą dobę przysługuje dieta w pełnej wysokości, do 8 godzin przysługuje 1/3 diety, 8-12 godzin przysługuje 1/2 diety, ponad 12 godzin przysługuje pełna dieta.

Kursy walut znajdują się w pliku tekstowym pobranym z Internetu, przed użyciem trzeba dokonać konwersji na plik xls. Podobnie, dane o dietach zostały pobrane z Internetu – wprowadzić w formie skoroszytu, ale także będzie konieczna konwersja na formę umożliwiającą obliczenia.

Po wyliczeniach w bazie delegacji wykonasz tabelę podsumowującą wypłaty w podziale na grupę delegacji krajowych i zagranicznych.

Po dokonaniu obliczeń musisz wysłać tabelę emailem do rewidenta celem sprawdzenia. Przed wysłaniem trzeba będzie zlikwidować łącza do arkuszy z dietami i kursami walut.

NARZĘDZIA:

1. Import plików tekstowych, kreator podziału tekstu na kolumny.
2. Formaty niestandardowe dotyczące dat.
3. Wykorzystanie funkcji PODAJ.POZYCJĘ oraz INDEKS do wyszukiwania danych z tabel według wierszy oraz kolumn.
4. Funkcje LICZBA.CAŁK, LITERY.MAŁE, ZAOKR, ZNAJDŹ, ILE.LICZB.
5. Budowanie formuł korzystających z innych plików, przerywanie łącz.
6. Funkcja sprawdzająca wystąpienie błędu CZY.BŁĄD.
7. Funkcje warunkowe SUMA.JEŻELI, LICZ.JEŻELI i ŚREDNIA.JEŻELI.

Wyszukaj plików
1 - co szukam
2 - w jakim zakresie
3 - co ma być (kolumna)
4 - o

ETAPY BUDOWY ROZWIĄZANIA


KROK 1

Obliczenie kwoty zwrotu kosztów przejazdu.

1. Otwórz plik **Rozliczenie delegacji**, zwróć uwagę, na dane znajdujące się na arkuszu **Delegacje** oraz **Dane pomocnicze**.
2. Na arkuszu **Dane pomocnicze** nazwij następujące obszary (najlepiej za pomocą pola nazwy):
 - 2.1. **trasy A2:B17**
 - 2.2. **dieta_krajowa B20**
 - 2.3. **stawka_km B21**
3. Na arkuszu Delegacje dodaj nagłówki dwóch nowych kolumn:
 - 3.1. **Ilość kilometrów (G3)**
 - 3.2. **Zwrot kosztów przejazdu (H3)**
4. W komórce G4 wstaw za pomocą polecenia Wstaw funkcję lub wpisz funkcję pobierającą liczbę kilometrów. Najlepiej do tego celu nadaje się funkcja WYSZUKAJ.PIONOWO. Ponieważ trasy podane są „w jedną stronę”, mnożymy wynik razy 2.

=WYSZUKAJ.PIONOWO(F4;trasy;2;FAŁSZ)*2

5. Pamiętaj, że aby wkleić nazwę obszaru (komórki) można posłużyć się przyciskiem **F3**.



6. Skopiuj formułę do końca tabeli.
7. W komórce H4 oblicz kwotę zwrotu kosztów przejazdu:

=G4*stawka_km

8. Skopiuj formułę do końca tabeli.

KROK 2

Obliczenie ilości diet należnych pracownikowi.

9. Na arkuszu Delegacje dodaj nagłówki dwóch nowych kolumn:

9.1. **Czas pobytu** (I3)

9.2. **Ilość diet** (J3)

10. Wpisz w komórce I4 formułę obliczającą czasu pobytu:


=C4-B4

11. Zwróć uwagę na format – został on nadany automatycznie, niestety jest nieprawidłowy. Zmień format na format ogólny (Rozwiń listę **Format liczb** z karty **Narzędzia główne** i wybierz pozycję **Ogólne**).



12. Skopiuj formułę do końca tabeli.

13. Format jest bardziej czytelny, niemniej jednak gdyby tabela miała być drukowana, lepszy byłby format prezentujący czas pobytu w podziale na dni oraz godziny. Niestety takiego formatu nie ma standardowo, można jednak użyć formatu niestandardowego.
14. Zaznacz obszar z czasem pobytu, a następnie kliknij prawym przyciskiem myszki na zaznaczeniu, z menu podręcznego wybierz polecenie **Formatuj komórki**, przejdź na zakładkę **Liczby** i wybierz kategorię **Niestandardowe**.
15. W polu **Typ** wpisz format: **d "dni" gg "godzin"**



16. Litera **d** jest kodem pokazującym liczbę dni, litery **gg** pokazują liczbę godzin (zawsze dwucyfrowo), teksty w cudzysłowach pojawiają się w odpowiednich miejscach pomiędzy liczbami.
17. „Przy okazji” możemy sformatować kolumnę G, zawierającą liczbę kilometrów. Zaznacz komórki **G4:G59**, a następnie kliknij prawym przyciskiem myszki na zaznaczeniu, z menu podręcznego wybierz polecenie **Formatuj komórki**, przejdź na zakładkę **Liczby** i wybierz kategorię **Niestandardowe**.
18. W polu **Typ** wpisz format: **0" km"**
19. Cyfra **0** jest kodem pokazującym liczbę, tekst w cudzysłowach jest „doklejony” do liczby (ale tylko na formacie, wartość pozostaje liczbą).

20. Następnie obliczymy ilość diet przysługujących za dany czas pobytu, zaznacz komórkę **J4** i wpisz (lub wybierz za pomocą polecenia Wklej funkcję) funkcję **JEŻELI** (jest ona zagnieżdżona ze względu na logikę obliczania przysługującej diety). Przy pisaniu formuły możesz posługiwać się poleceniami **Kopiuj** oraz **Wklej**. Funkcja **LICZBA.CAŁK** obcina część ułamkową liczby.

```
=LICZBA.CAŁK(I4)+JEŻELI(I4-LICZBA.CAŁK(I4)>12/24;1;  
JEŻELI(I4-LICZBA.CAŁK(I4)>8/24;0,5;  
JEŻELI(I4-LICZBA.CAŁK(I4)>0;1/3;0)))
```

21. Zmień format komórki na liczbowy (zaokrąglenie do jednego miejsca po przecinku) i skopiuj formułę do końca tabeli. Uzyskaliśmy poprawną liczbę diet dla każdej delegacji.

KROK 3

Obliczenie diety krajowej – ilość diet pomnożona przez stawkę dzienną diety krajowej. Wyznacznikiem czy wyjazd jest krajowy, czy zagraniczny jest słowo „Polska” umieszczone w kolumnie Kraj.

22. Do obliczenia diety krajowej posłużymy się funkcją **JEŻELI**, sprawdzając czy w kolumnie Kraj jest słowo „Polska”. Ponieważ porównywanie tekstów jest literalne (z dokładnością do małych i dużych liter), zaś słowo może być pisane różnie (na przykład ze spacją), do sprawdzenia posłużymy się funkcją **ZNAJDŹ**. Funkcja ta wyszukuje tekst będący pierwszym argumentem w tekście (komórce) będącym drugim argumentem.
23. Wpisz tytuł kolumny **Dieta krajowa** w komórce **K3**.
24. Zaznacz komórkę **K4** i wpisz do niej formułę:

```
=ZNAJDŹ("polska";LITERY.MAŁE(E4))
```


25. Aby uniknąć problemów z analizą pisowni, szukamy słowa „polska” (pisanego małymi literami), zaś wartość komórki E4 zamieniamy na małe litery za pomocą funkcji **LITERY.MAŁE**.
26. Zwróć uwagę, że funkcja zwraca kod błędu (**#ARG!**) jeżeli nie znajdzie słowa „polska”, w przeciwnym wypadku zwracana jest liczba **1** – numer znaku, w którym zaczyna się słowo w komórce.

Teraz możemy uzupełnić formułę o funkcję **JEŻELI**. Do sprawdzenia, czy w komórce jest błąd zostanie użyta funkcja **CZY.BŁĄD**, natomiast do zaokrąglenia wartości diety do dwóch miejsc po przecinku użyjemy funkcji **ZAOKR**:

```
=JEŻELI(CZY.BŁĄD(ZNAJDŹ("polska";LITERY.MAŁE(E4)));0;  
ZAOKR(J4*dieta_krajowa;2))
```

27. Skopiuj formułę do końca tabeli i sformatuj ją w zapisie walutowym.


28. Dla wyjazdów zagranicznych pokazywana jest wartość 0, jeżeli nie chcemy wyświetlać wartości zerowych należy kliknąć **Przycisk pakietu Office** następnie kliknąć przycisk **Opcję programu Excel**, przejść na zakładkę **Zaawansowane** i odznaczyć opcję **Pokaż zero w komórkach o zerowej wartości**.




KROK 4

Podobnie jak przy diecie krajowej, należy obliczyć dietę zagraniczną. Dane na temat kwot (oraz walut) dla poszczególnych krajów umieszczone są na osobnym arkuszu, który trzeba będzie najpierw przygotować, aby mógł posłużyć jako źródło danych.

29. Na arkuszu Delegacje dodaj nagłówki dwóch nowych kolumn:
- 29.1. **Dieta zagraniczna** (L3)
 - 29.2. **Waluta** (M3)
 - 29.3. **Kurs** (N3)
 - 29.4. **Dieta zagraniczna w PLN** (O3)
30. Otwórz plik **Diety** i zwróć uwagę, w jakiej postaci są dane, zawarte w pliku. Zostały one zaimportowane z sieci Internet, całość informacji jest w istocie zawarta w kolumnie A.
31. Do podziału danych na kolumny użyjemy Kreatora podziału tekstu na kolumny.
32. Usuń pierwszy wiersz.
33. Zaznacz komórki **A1:A95**.
34. Użyj polecenia **Tekst jako kolumny** z karty **Dane** z grupy **Narzędzia danych**.
35. Teksty są rozdzielone znakami spacji, zostaw więc opcję **Rozdzielany** w pierwszym kroku Kreatora i przejdź dalej za pomocą przycisku **Dalej**.
36. W następnym kroku Kreatora ustaw **Spację** jako znak rozdzielający, ustaw także Kwalifikator tekstu na „**brak**” (nazwy krajów nie są w cudzysłowach). Sprawdź, że pomiędzy kolumnami pojawiły się linie, co świadczy, że podział na kolumny przebiegnie prawidłowo. Potwierdź przyciskiem **Dalej**.



37. W trzecim kroku Kreatora zaznacz opcję **Nie importuj kolumny (pomiń)** dla pierwszej oraz piątej kolumny (nie będziemy potrzebować tych danych), a następnie zmień miejsce docelowe na **B1** (nie chcemy nadpisać danych źródłowych). Potwierdź przyciskiem **Zakończ**.



38. Sprawdź czy dane na arkuszu są poprawne. Dieta dla Japonii ma kropkę jako symbol części tysięcznej – usuń ją, tak aby w komórce pojawiła się liczba.
39. Usuń kolumnę **A**, dodaj wiersz nad danymi, wpisz do niego nagłówki: **Kraj, Waluta, Dieta**. Arkusz zawiera teraz prawidłowo przygotowane dane.
40. Przejdź do pliku **Rozliczenie delegacji** (arkusz Delegacje), zaznacz komórkę **L4**.
41. Dietę zagraniczną pobierzemy teraz za pomocą funkcji WYSZUKAJ.PIONOWO. Użyj polecenia Wstaw funkcję (lub wpisz funkcję do komórki) – obszar w pliku diety zaznacz myszką:

```
=WYSZUKAJ.PIONOWO(E4;'[Diety.xls]Kwoty  
diet'!$A$2:$C$96;3;FAŁSZ)
```

42. Zwróć uwagę, na formę odwołania do innego pliku: nazwa pliku w nawiasach kwadratowych, nazwa arkusza, wykrzyknik, zakres komórek. Nazwa pliku i arkusza jest w pojedynczych cudzysłowach.
43. Po skopiowaniu formuły zauważysz, że dla delegacji krajowych pojawił się błąd (kraju Polska nie ma na liście krajów. Przejdź z powrotem do komórki L4 i uzupełnij formułę o funkcję JEŻELI:

```
=JEŻELI(K4=0;WYSZUKAJ.PIONOWO(E4;'[Diety.xls]Kwoty  
diet'!$A$2:$C$96;3;FAŁSZ);0)
```

44. Skopiuj formułę do końca tabeli.
45. W podobny sposób pobierzemy nazwy walut (będą potrzebne do przeliczenia ich na złote w następnym kroku). Przejdź do komórki **M4**, a następnie za pomocą polecenia Wstaw funkcję (lub wpisując bezpośrednio) dodaj następującą formułę:

```
=JEŻELI(K4=0;WYSZUKAJ.PIONOWO(E4;'[Diety.xls]Kwoty  
diet'!$A$2:$C$96;2;FAŁSZ);0)
```

46. Skopiuj formułę do końca tabeli.
47. Przejdź do pliku **Diety** i zamknij go (zapisując zmiany).
48. Zwróć uwagę, że formuły w kolumnie **L** i **M** zawierają teraz pełne ścieżki do pliku Diety.


KROK 5

Korzystając z symbolu waluty, dokonamy przeliczenia diet zagranicznych na PLN po kursie z dnia wyjazdu. Dane na temat kursów walut w okresie objętym delegacjami otrzymaliśmy w postaci pliku tekstowego. Będziemy go musieli najpierw zaimportować.


49. Użyj polecenia **Otwórz** klikając **Przycisk pakietu Office**, a następnie wybierz na liście Pliki typu (na dole okna) pozycję **Pliki tekstowe**.
50. Wyszukaj plik **Kursy2010** i potwierdź przyciskiem **Otwórz**.
51. Na ekranie pojawi się Kreator importu tekstów, przejdź do następnego kroku za pomocą przycisku **Dalej**.
52. W drugim kroku Kreatora włącz **Tabulator** jako ogranicznik, przejdź do następnego kroku za pomocą przycisku **Dalej**.




53. Zaznacz **pierwszą kolumnę** i wybierz dla niej opcję **Data: RMD** (spowoduje to, że kreator prawidłowo rozpozna datę).



54. Następnie użyj przycisku **Zaawansowane** i wybierz **kropkę** jako separator dziesiętny (dane w pliku zawierają kropki zamiast przecinków). Potwierdź przyciskiem **OK**.



55. Potwierdź Kreatora przyciskiem **Zakończ** i sprawdź czy plik zawiera prawidłowe dane (daty w pierwszej kolumnie, liczby w następnych kolumnach).
56. Użyj polecenia **Zapisz Jako** klikając **Przycisk pakietu Office**, wybierz typ pliku **Skoroszyt programu Excel** i potwierdź operację przyciskiem **Zapisz** (jako nazwę pliku pozostaw **Kursy 2010**). Uwaga – gdybyśmy tego nie uczynili, w dalszym ciągu pracowalibyśmy na pliku tekstowym!



57. Plik posłuży nam do pobrania właściwych kursów.
58. Przejdź do pliku **Rozliczenie delegacji**, zaznacz komórkę **N4**.
59. Do pobrania kursu posłużymy się kombinacją funkcji **PODAJ.POZYCJĘ** oraz **INDEKS** (musimy odnaleźć kurs dla nazwy waluty jak też i prawidłowej daty). Funkcja **INDEKS** zwraca wartość na przecięciu podanego numeru wiersza i kolumny z wybranego obszaru, numery te pobierzemy za pomocą funkcji **PODAJ.POZYCJĘ**.
60. Za pomocą polecenia **Wstaw funkcję**, wstaw funkcję **INDEKS (postać tablicowa)**, w pierwszym argumencie funkcji (**Tablica**) zaznacz za pomocą myszki całą tabelę na pliku **Kursy2010 (A1:N170)**.
61. W argumencie **Nr.kolumny** wstaw funkcję **PODAJ.POZYCJĘ** (za pomocą listy rozwijalnej w lewym górnym rogu). Funkcja zwróci numer kolumny, w której znajduje się wybrana waluta.

PODAJ.POZYCJE	
Szukana_wartość	M4
Przeszukiwana_tab	[zanie).xlsx]Kursy2010!\$A\$1:\$N\$1
Typ_porównania	FALSZ

62. W argumencie **Nr.wiersza** wstaw funkcję **PODAJ.POZYCJĘ** (za pomocą listy rozwijalnej w lewym górnym rogu). Funkcja zwróci numer wiersza, w której znajduje się data **równa lub mniejsza od wybranej daty wyjazdu** (stąd trzeci argument jest równy prawdzie).

PODAJ.POZYCJE	
Szukana_wartość	B4
Przeszukiwana_tab	[nie).xlsx]Kursy2010!\$A\$1:\$A\$170
Typ_porównania	PRAWDA

63. Potwierdź Kreatora funkcji za pomocą przycisku **OK** i sprawdź czy w komórce N4 znajduje się prawidłowy kurs. Komórka ta powinna zawierać następującą formułę:

```
=INDEKS('[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$N$170;PODAJ.POZYCJE(B4;'[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$A$170;PRAWDA);PODAJ.POZYCJE(M4;'[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$N$1;FALSZ))
```

64. Uzupełnij formułę o funkcję JEŻELI (kursy nie powinny być pobierane dla delegacji krajowych).


```
=JEŻELI(M4=0;0;INDEKS('[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$N$170;PODAJ.POZYCJE(B4;'[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$A$170;PRAWDA);PODAJ.POZYCJE(M4;'[Kursy2010(rozwiazanie).xlsx]Kursy2010!$A$1:$N$1;FALSZ)))
```

65. Skopiuj formułę do końca tabeli.
66. Zaznacz komórkę O4 i wpisz formułę obliczającą dietę zagraniczną:
=JEŻELI(N4=0;0;ZAOKR(J4*L4*N4;2))
67. Skopiuj formułę do końca tabeli i sformatuj jako zapis walutowy.
68. Przejdź do pliku **Kursy2010** i zamknij go (zapisując zmiany).
69. Zwróć uwagę, że formuły w kolumnie **N** zawierają teraz pełne ścieżki do pliku Kursy2010.
70. Zaznacz kolumny **L:M** i użyj polecenia **Grupuj** z karty **Dane** z grupy **Konspekt**.

KROK 6

Po dokonaniu obliczeń dla poszczególnych delegacji wykonasz tabelę podsumowującą wszystkie delegacje. Wykonaj podsumowanie delegacji krajowych za pomocą funkcji **SUMA.JEŻELI**, **LICZ.JEŻELI** oraz **ŚREDNIA.JEŻELI** z wyszczególnieniem delegacji, których koszt wynosi powyżej 1000 PLN a następnie delegacji zagranicznych w rozbiciu na użyte waluty (EUR i USD).

71. W komórce B2 uruchom funkcję **LICZ.JEŻELI**. Bieżącym **Kryterium** będzie tekst Polska, natomiast **Zakresem** komórek będzie cała kolumna z bazy danych Delegacje, w której znajdują się nazwy krajów.



Ze względu na to, że funkcja będzie później kopiowana zablokuj Zakres. Możesz użyć do tego klawisza F4.

LICZ.JEŻELI()

Funkcja potrafi obliczyć ile razy komórka spełniająca podane kryterium pojawia się we wskazanym obszarze.
Przykład podany w ćwiczeniu: ile razy w kolumnie Kraj pojawia się komórka z wpisem Polska?

Funkcja LICZ.JEŻELI ma dwa argumenty:
Zakres – kolumna w bazie danych, w ramach której chcesz odnaleźć bieżące kryterium (kolumna Kraj)
Kryterium – określenie jakich komórek szukamy (Polska).

72. W komórce C2 wstaw funkcję **SUMA.JEŻELI**. Ta funkcja będzie miała trzy okna argumentów – oprócz Zakresu i Kryterium należy podać jeszcze zakres komórek do sumowania.

SUMA.JEŻELI() oraz ŚREDNIA.JEŻELI()

Funkcje potrafią wyliczyć sumę (średnią) tylko dla komórek spełniających podane kryterium.

Przykład podany w ćwiczeniu: jaka jest suma wypłat dla delegacji krajowych?

Funkcja SUMA.JEŻELI / ŚREDNIA.JEŻELI ma trzy argumenty:
Zakres – obszar przeszukiwany - kolumna w bazie danych, w ramach której chcesz odnaleźć bieżące kryterium (kolumna Kraj)

Kryterium – określenie jakich komórek szukamy (Polska)

Suma_zakres – obszar komórek, które mają być sumowane jeśli kryterium będzie zgodne - kolumna w bazie danych, dla której wykonujesz podsumowanie (kolumna Do wypłaty).

SUMA.JEŻELI		
Zakres	Delegacje!\$E\$4:\$E\$260	= {"Litwa","Polska","Niemcy","Litwa"}...
Kryteria	"polska"	= "polska"
Suma_zakres	Delegacje!\$P\$4:\$P\$260	= {1062,97 204,38 1366,51 901,03...}
		= 38654,405

73. Skopiuj funkcję do komórki **D2**.
Zmień użytą funkcję na funkcję **ŚREDNIA.JEŻELI**. W tym celu stań w pasku formuły na nazwie funkcji, następnie z listy rozwijalnej funkcji wybierz Więcej funkcji i następnie **ŚREDNIA.JEŻELI**. Zatwierdź funkcję.


=ŚREDNIA.JEŻELI(Delegacje!\$E\$4:\$E\$260;"polska";Delegacje!\$P\$4:\$P\$260)

74. Przejdź do komórki **B4**. W tej komórce musisz podać liczbę delegacji krajowych z kosztem powyżej 1000 PLN. W tym celu użyjesz formuły **BD.ILE.REKORDÓW**. Funkcja ta należy do grupy funkcji baz danych (są omawiane szczegółowo w ćwiczeniu Artykuły sportowe 2). Kryteria dla funkcji zostały przygotowane w komórkach H2:I3. Wpisz w komórce formułę:

=BD.ILE.REKORDÓW(Delegacje!\$A\$3:\$P\$260;"Do wypłaty";H2:I3)

75. Przejdź do komórki **B5**. Tutaj należy zliczyć pozycje dotyczące wypłat z tytułu delegacji zagranicznych. Aby uzyskać wynik, zlicz wszystkie wypłaty i odejmij od nich liczbę delegacji krajowych.

=ILE.LICZB(Delegacje!P4:P260)-'Do wypłaty'!B2



76. W komórce **C5** analogicznie oblicz sumę wypłat z tytułu delegacji zagranicznych.
- =SUMA(Delegacje!P4:P260)-'Do wypłaty'!C2**
77. W komórce **D5** podziel sumę wypłat przez liczbę delegacji : **=C5/B5**.
78. Wiersz piąty uzupełnij funkcjami **LICZ.JEŻELI**, **SUMA.JEŻELI** oraz **ŚREDNIA.JEŻELI**. Skopiuj formułę z **B2** do **B7**. Następnie stojąc w **B7** kliknij na znak **fx**. W oknie **Kryteriów** funkcji zmień odwołanie na **A7**, a w oknie **Zakres** zaznacz kolumnę z walutami w bazie delegacji. Podobnie skopiuj i zmodyfikuj funkcje **SUMA.JEŻELI** i **ŚREDNIA.JEŻELI**.

komórka B7:

=LICZ.JEŻELI(Delegacje!\$M\$4:\$M\$260;\$A\$7)


komórka C7:

=SUMA.JEŻELI(Delegacje!\$M\$4:\$M\$260;\$A\$7;Delegacje!\$P\$4:\$P\$260)

komórka D7:

=ŚREDNIA.JEŻELI(Delegacje!\$M\$4:\$M\$260;\$A\$7;Delegacje!\$P\$4:\$P\$260)

79. W komórce **A7** masz mieć możliwość wyboru walut. Źródło listy znajduje się w komórkach **H7:H12**.
80. Przejdź do **A7** i wybierz polecenie **Poprawność danych** z menu **Dane**. Z listy **Dozwolone** wybierz **Lista** i jako źródło zaznacz komórki z nazwami walut. Potwierdź OK.




81. Ukryj kolumny z nazwami walut i kryteriami dla delegacji krajowych i zapisz plik.

Podaj funkcję

KROK 7

Zamierzasz wysłać email'em rozliczenie delegacji do sprawdzenia. Osoba przeglądająca delegacje nie będzie posiadała dostępu do plików z dietami zagranicznymi oraz kursami walut.

83. Zamknij plik **Rozliczenie delegacji** (zapisując zmiany), a następnie otwórz go ponownie.
84. Na ekranie pojawi się komunikat o aktualizowaniu łączy. Wybierz opcję Aktualizuj – gdyby dane w plikach źródłowych zmieniły się, zostałyby one automatycznie zmienione w formułach.
85. Zapiszemy plik pod inną nazwą i zamienimy łączy na wartości.
86. Użyj polecenia **Zapisz Jako** klikając **Przycisk Pakietu Office**, ustaw typ pliku na **Skoroszyt programu Excel**, jako nazwę pliku wpisz **Rozliczenie delegacji wersja 2** a następnie kliknij **Zapisz**.
87. Użyj polecenia **Edytuj łączy** z karty **Dane** z grupy **Połączenia**, na ekranie pojawi się okno z łączy do dwóch plików.



88. Zaznacz kolejno każde łączy na liście i użyj przycisku **Przerwij łączy**.
89. Potwierdź komunikaty ostrzegawcze, zamknij okno Edycji łączy i sprawdź, że na arkuszu kursy i nazwy walut a także kwoty diet zagranicznych są wartościami a nie formułami tak jak poprzednio.
90. Taki plik wyślemy email'em bez obawy o problemy z aktualizacją łączy.

ROZWIĄZANIE

Arkusz Delegacje

	H	T	J	K	L	M	N	O	P
3	Zwrot kosztów przejazdu	Czas pobytu	Ilość diet	Dieta krajowa	Dieta zagraniczna	Waluta	Kurs	Dieta zagraniczna w PLN	Do wypłaty
4	1 253,70 PLN	3 dni 12 godzin	3,5			33 EUR	4,1084	474,52 PLN	1 728,22 PLN
5	501,48 PLN	3 dni 00 godzin	3,0	69,00 PLN					570,48 PLN
6	1 387,28 PLN	4 dni 06 godzin	4,3			42 EUR	4,0595	728,83 PLN	2 076,11 PLN
7	1 253,70 PLN	2 dni 01 godzin	2,3			33 EUR	4,0595	312,58 PLN	1 566,28 PLN
8	501,48 PLN	2 dni 06 godzin	2,3	53,67 PLN					555,15 PLN
9	1 337,28 PLN	3 dni 12 godzin	3,5			42 EUR	4,0595	596,75 PLN	1 934,03 PLN
10	501,48 PLN	4 dni 06 godzin	4,3	99,67 PLN					601,15 PLN
11	1 337,28 PLN	3 dni 00 godzin	3,0			42 EUR	4,0595	511,50 PLN	1 848,78 PLN
12	1 337,28 PLN	1 dni 12 godzin	1,5			33 EUR	4,0595	200,95 PLN	1 538,23 PLN
13	1 337,28 PLN	5 dni 03 godzin	5,3			33 EUR	4,0595	714,47 PLN	2 051,75 PLN
14	1 387,28 PLN	5 dni 03 godzin	5,3			42 EUR	4,0814	914,23 PLN	2 281,51 PLN
15	1 002,96 PLN	1 dni 12 godzin	1,5			42 EUR	4,0814	257,13 PLN	1 260,09 PLN
16	1 002,96 PLN	2 dni 00 godzin	2,0			42 EUR	4,0814	342,84 PLN	1 345,80 PLN
17	1 002,96 PLN	4 dni 06 godzin	4,3			42 EUR	4,0814	742,81 PLN	1 745,77 PLN
18	501,48 PLN	3 dni 00 godzin	3,0	69,00 PLN					570,48 PLN
19	501,48 PLN	3 dni 11 godzin	3,5	80,50 PLN					581,98 PLN
20	1 253,70 PLN	1 dni 12 godzin	1,5			33 EUR	4,0615	201,04 PLN	1 454,74 PLN
21	501,48 PLN	5 dni 03 godzin	5,3	122,67 PLN					624,15 PLN
22	1 337,28 PLN	3 dni 12 godzin	3,5			42 EUR	4,0594	595,85 PLN	1 933,13 PLN

Arkusz Do wypłaty

	A	B	C	D
1	Rodzaj delegacji	Liczba delegacji	Suma w PLN	Średnia w PLN
2	Krajowe	105 SZT.	58 654,41 PLN	558,61 PLN
3	w tym dotyczy kosztu powyżej (podaj)			
4	1 000,00 PLN	4 SZT.		
5	Zagraniczne	152 SZT.	372 783,10 PLN	2 452,52 PLN
6	w tym dotyczy waluty (wybierz z listy)			
7	USD	20 SZT.	50 272,88 PLN	2 513,64 PLN
8	USD			
9	EUR			
10	CHF			
11	GBP			
12	SKK			
13	RUB			

Jeżeli:
 a) cos = "cos" (nie wiem)
 2) to
 3) 0

Przy kopiowaniu posłuchać z posłuchać
 2) Kosz. posłuchać -> Copy -> ESC i skopiuj
 posłuchać posłuchać
 upiniję.

Chciał N →